

Nameless, Painless

Nicolas Pouillard

INRIA

Nicolas.Pouillard@inria.fr

Abstract

De Bruijn indices are a well known technique for programming with names and binders. They provide a representation that is both simple and canonical.

However, programming errors tend to be really easy to make. We propose a safer programming interface implemented as a library. Whereas indexing the types of names and terms by a numerical bound is a famous technique, we index them by worlds, a different notion of index that is both finer and more abstract. While being more finely typed, our approach incurs no loss of expressiveness or efficiency.

Via parametricity we obtain properties about functions polymorphic on worlds. For instance, well-typed world-polymorphic functions over open λ -terms commute with any renaming of the free variables.

Our whole development is conducted within Agda, from the code of the library, to its soundness proof and the properties of external functions. The soundness of our library is demonstrated via the construction of a logical relations argument.

Categories and Subject Descriptors D.3.3 [Programming Languages]: Language Constructs and Features—Data types and structures; Polymorphism

General Terms Design, Languages, Theory

Keywords names, binders, meta-programming, name abstraction, de Bruijn indices

1. Introduction

It is quite common in the programming realm to deal with the mundane business of data structures with names and binders. Compilers, code generators, static analysers, theorem provers, and type-checkers have this in common. They manipulate programs, formulae, proofs, and types. When seen as pieces of data, there is a common difficulty: the representation of variables (names and binders).

One traditional approach is to represent all the occurrences of a bound variable identically by using character strings or integers. While being the most obvious representation it is known to cause a lot of trouble when dealing with operations like substitution. In particular we name it the capture-avoiding substitution because it has to rename some variables to avoid accidental changes called captures.

In this article we focus on a different kind of representation, namely de Bruijn indices [6]. This representation is said to be nameless because variables are no longer identified by a name but a notion of “distance” to the binding point. De Bruijn indices are the most famous nameless representation and in the following we will use the term “nameless” as a synonym for “de Bruijn indices”.

The nameless approach solves part of the problem by providing a canonical representation. More precisely binding occurrences are no longer named (we now use λ . instead of $\lambda x.$). Then bound variables are represented by the “distance” to the binding λ . This distance is the number of λ s to cross in order to reach the binding λ .

A major issue with this nameless representation is its arithmetic flavor. Indeed properties about names and binders are turned into arithmetic formulae. The result can be harder to understand, and worse, easier to get wrong. To remedy this, several techniques have been proposed to give finer grained types to these representations [1, 2, 5, 7]. We continue further in this direction by providing a safer system to program with a nameless representation.

1.1 Contributions

In a previous paper [10] we developed an abstract interface to program with names and binders. The interface can be both implemented in a named or a nameless style. Here by focusing on a nameless representation we greatly simplify the programming interface needed to achieve our safety goals. This is at the expense of some loss of abstraction. However while the goals are less ambitious, the results are significantly more applicable since simpler and closer to a concrete representation.

Our previous nameless implementation was relying on the `Fin` approach (described in section 1.4.3). The notion of worlds presented here (in section 2.3) is significantly more precise than `Fin`. However the worlds were already abstract to the programmer and they still are. Indeed abstraction is necessary to retain the parametricity properties we want from world-polymorphic functions.

While a fair amount of proofs were already mechanized in [10], the logical relation argument was still largely on paper. Here we not only finished these mechanized proofs but have every part well connected in one single setup. The code and proofs are accessible online at [9] and we encourage the reader to have a look for reference on technical details and applications.

1.2 Outline of the paper

This paper is organized as follows. In the remainder of this section we present several nameless techniques to represent data structures with bindings. In section 2 we detail our approach to the problem. In section 3 we give some examples and advanced operations. In section 4 we develop a logical relation construction which ensures the soundness properties we want, and we show how to exploit these properties on concrete examples. All the construction is built in the AGDA language, our system is an AGDA library and its soundness proof is done in AGDA as well.

Permission to make digital or hard copies of all or part of this work for personal or classroom use is granted without fee provided that copies are not made or distributed for profit or commercial advantage and that copies bear this notice and the full citation on the first page. To copy otherwise, to republish, to post on servers or to redistribute to lists, requires prior specific permission and/or a fee.

ICFP'11, September 19–21, 2011, Tokyo, Japan.

Copyright © 2011 ACM 978-1-4503-0865-6/11/09...\$10.00

1.3 A brief introduction to AGDA notation

Throughout the paper, our definitions are presented in the syntax of AGDA. In AGDA, `Set` (or `Set0`) is the type of small types like `Bool`, `Maybe (List Bool)`, or `ℕ`. `Set1` is the type of `Set`, `Set → Bool`, or `ℕ → Set`. The function space is written `A → B`, while the dependent function space is written `(x : A) → B` or `∀ (x : A) → B`. An implicit parameter, introduced via `∀ {x : A} → B`, can be omitted at a call site if its value can be inferred from the context. There are shortcuts for introducing multiple arguments at once or for omitting a type annotation, as in `∀ {A} {i j : A} x → e`. There is no specific sort for propositions in AGDA, everything is in `Set ℓ` for some `ℓ`. The unit type is a record type with no fields named `⊤`, it also represents the `True` proposition. The empty type is an (inductive) data type with no constructors named `⊥`, it also represents the `False` proposition. The negation `¬ A` is defined as `A → ⊥`.

The same name can be used for different data constructors. AGDA makes use of type annotations to resolve ambiguities.

As in Haskell, a definition consists of a type signature and a sequence of defining equations, which may involve pattern-matching. The `with` construct extends a pattern-matching-based definition with new columns. An ellipsis `...` is used to elide a redundant equation prefix.

AGDA is strict about whitespace: `x≤y` is an identifier, whereas `x ≤ y` is an application. This allows naming a variable after its type (deprived of any whitespace). We use mixfix declarations, such as `_⊕_`. We use some definitions from AGDA’s standard library: natural numbers, booleans, lists, and applicative functors (`pure`, `_⊗_`).

For the sake of conciseness, the code fragments presented in the paper are sometimes not perfectly self-contained. However, a complete AGDA development is available online [9].

1.4 Related work: nameless representations

Various techniques have been discovered to build a nameless representation. We have chosen a few of them which gradually setup the stage.

1.4.1 bare: The original approach

We call this one `bare`, because it solely relies on natural numbers. To make things more concrete here is an example of its use when defining the untyped λ -calculus with local bindings (`let`). Our naming convention is as follows: we call `Tm` the type of λ -terms, `V` is the data constructor for variables, `_·_` is for the application, `λ` is for the λ -abstraction, and finally `Let` is for local bindings.

```
data Tm : Set where
  V      : (x : ℕ) → Tm
  _·_    : (t u : Tm) → Tm
  λ      : (t : Tm) → Tm
  Let    : (t u : Tm) → Tm
```

From the point of view of the binding structure, it is striking that no difference appears between the constructors of the data type. It is completely up to the programmer to manage the scoping difference introduced by `λ` and `Let`. This is even more worrying in the `Let` case since we have no clue of difference between the arguments.

Here is an example using this representation to build the λ -term for `λf . λx . f x`.

```
appTm : Tm
appTm = λ (λ (V 1 · V 0))
```

The main advantages of this approach are its simplicity and its expressiveness. The expressiveness is almost maximal since no

restriction is put on the usage of variables. Actually expressiveness is really maximal only in partial languages where all types are inhabited by a looping term. Indeed the absence of restrictions on data can be constraining when receiving data as argument. In a total language all cases have to be covered, even those we consider as impossible. For instance a function accepting only closed terms will have to provide a return value for the variable case anyway. Sometimes the return type is so constrained that the type is actually empty in this case, and so no dummy value can be returned. However, even in a partial language the expressiveness is maximal if we are willing to live in an error monad.

1.4.2 Maybe: The nested data type approach

The nested data type approach [2, 3, 5] is a first step towards better properties about the binding structure of terms. Let us start with the definition of `Tm` with this approach:

```
data Tm (A : Set) : Set where
  V      : (x : A) → Tm A
  _·_    : (t u : Tm A) → Tm A
  λ      : (t : Tm (Maybe A)) → Tm A
  Let    : (t : Tm A) (u : Tm (Maybe A)) → Tm A
```

There are three points to look at. The `Tm` type is parameterized by another type called `A`, so we can look at it as a kind of container. Note also that the variable case `V` does not hold a `ℕ` but an `A`. Last but not least the `λ` case holds a term whose parameter is not simply an `A` but a `Maybe A`.

This last point makes the `Tm` type a nested data type, also called a non-regular type. This has the consequence of requiring polymorphic recursion to write recursive functions on such a type.

To understand why this is an adequate representation of λ -terms one has to look a bit more at the meaning of `Maybe`. If types are seen as sets, then `Maybe` takes a set and returns a set with one extra element. So each time we cross a `λ`, there is one extra element in the set of allowed variables, exactly capturing the fact we are introducing a variable.

To see the difference with the previous approach, here is the `appTm` λ -term again:

```
appTm : Tm ⊥
appTm = λ (λ (V (just nothing) · V nothing))
```

Note the use of the empty type `⊥` to state the closedness of the `appTm` term. Stating this kind of properties was impossible to do with the previous approach, without resorting to logical properties on the side.

1.4.3 The Fin approach

Another approach already described and used in [1, 7] is to index everything (terms for example) by a bound. This bound is the maximum number of distinct free variables allowed in the value. This rule is enforced in two parts: variables have to be strictly lower than their bound, and the bound is incremented by one when crossing a name abstraction (a λ -abstraction for instance, called `λ` here).

The `Fin n` type is used for variables and represents natural numbers strictly lower than `n`. The name `Fin n` comes from the fact that it defines finite sets of size `n`. We call this approach `Fin` for its use of this type. The definition found in AGDA standard library is the following:

```

data Fin : ℕ → Set where
  zero : {n : ℕ} → Fin (suc n)
  suc  : {n : ℕ} (i : Fin n) → Fin (suc n)

```

Given this `Fin` type, one can define the `Tm` data type using this approach:

```

data Tm n : Set where
  V      : (x : Fin n)           → Tm n
  _'_   : (t u : Tm n)         → Tm n
  λ     : (t : Tm (suc n))     → Tm n
  Let   : (t : Tm n) (u : Tm (suc n)) → Tm n

```

As the previous approach, this representation helps enforcing some wellformedness properties, for instance `Tm 0` is the type of closed λ -terms.

Here is the `appTm` λ -term in this approach:

```

appTm : Tm 0
appTm = λ (λ (V (suc zero) · V zero))

```

We can easily draw a link with the `Maybe` approach. Indeed, the `Fin (suc n)` type has exactly one more element than the `Fin n` type. Although they are not equivalent for at least two reasons. The `Maybe` approach can accept any type to represent variables. This makes the structure more like a container and this can be particularly helpful to define the substitution as the composition of `mapTm` : $\forall \{A B\} \rightarrow (A \rightarrow B) \rightarrow Tm A \rightarrow Tm B$ and `joinTm` : $\forall \{A\} \rightarrow Tm (Tm A) \rightarrow Tm A$ as in [2, 5]. The `Fin` approach has advantages as well, the representation is concrete and simpler since closer to the `bare` approach. However this apparent simplicity comes at a cost, we will see that its concrete representation is one root of the problem.

2. NAPA, a safer nameless representation

2.1 Motivation

While these nameless approaches easily guarantee wellformedness of terms, nothing is said about functions manipulating terms. How are the inputs of a function related to its output? What should a function be able to do given unknown free variables? We claim that free variables should be kept abstract. More precisely a function f operating on terms should commute with a renaming Φ of free variables, namely $f \circ \Phi = \Phi \circ f$. By renaming we mean a permutation of the free variables (more precisely an injective function). In fact this is a bit more flexible, depending on the type of the function f this property will be enforced or not, as we will see in section 4.5.

The goal is to catch some programming errors by having finer types for the manipulated values and for the functions manipulating values. We will show later that this is not a restriction by itself since the types do not force the programmer to be precise everywhere.

One common kind of programming error with these nameless representations is to forget to shift the free variables here and there. While these errors are already caught by the `Maybe` and the `Fin` approaches, in these approaches the programmer is provided with means to workaround the type error. In complex situations making the distinction between a valid coercion and a wrong one can be challenging. With finer types, the mistakes will result in more informative type errors.

2.2 A data definition kit

The three previous approaches are closely related. Just a few details change each time. We can capture all of them with a single abstraction: a triple $(World, Name, _ \uparrow 1)$ where `World` is the index

```

bareKit  : DataKit
bareKit  = ⟨ T , λ _ → ℕ , id ⟩

```

```

maybeKit : DataKit
maybeKit = ⟨ Set , id , Maybe ⟩

```

```

finKit   : DataKit
finKit   = ⟨ ℕ , Fin , suc ⟩

```

Figure 1. DataKits for the three previous approaches

of types like `Tm`, `Name` is the type for names (indexed by worlds) as used in the `V` constructor, and finally `_ \uparrow 1` is an operation to shift a world by one as used in the `λ` constructor.

In AGDA, this triple can be defined with a record type named `DataKit`. This type is made universe polymorphic to accept the nested data types approach.

```

record DataKit {ℓ} : Set (suc ℓ) where
  constructor ⟨_,_,_⟩
  field
    World : Set ℓ
    Name  : World → Set
    _\uparrow 1 : World → World

```

The three previous approaches are summed up in figure 1 by defining their `DataKit`. Note that for the `bare` approach the index type is the unit type. In the `Maybe` approach the index type is `Set` and the type for names is directly the index so we use the identity function.

We can now give a single definition of `Tm` generalizing the three previous ones. To this end, we simply take a `DataKit` as argument and use it to define the type `Tm`.

```

module DataTm {ℓ} (kit : DataKit {ℓ}) where
  open DataKit kit
  data Tm α : Set ℓ where
    V      : (x : Name α)           → Tm α
    _'_   : (t u : Tm α)         → Tm α
    λ     : (t : Tm (α \uparrow 1)) → Tm α
    Let   : (t : Tm α) (u : Tm (α \uparrow 1)) → Tm α

```

We claim that nothing more is required from the kit. As complex binding structures can be defined in the original nameless representation (`bare`), they can also be defined using this kit. In particular having an empty world or making an induction over a world is not necessary.

Translation between data types using different kits can be done modularly as we do in the AGDA development. In any direction (any pair of kits $k_1 k_2$) one can define a function of type $\forall \{\alpha_1 \alpha_2\} \rightarrow (Name k_1 \alpha_1 \rightarrow Name k_2 \alpha_2) \rightarrow Tm k_1 \alpha_1 \rightarrow Tm k_2 \alpha_2$. This function takes a mapping for free variables and lifts this mapping to terms.

Whereas using these translations is fine in many cases, when going from `bare` to a more precise representation, an issue arises. Since AGDA is a total language, the initial mapping must be a total function. The `bare` approach ensuring nothing about free variables, one cannot supply the initial mapping to go to a closed term. One has to keep at least one free variable and then dynamically check for occurrences of the free variable in a second transformation.

We will now build on this `DataKit` abstraction to introduce our own solution.

```

World   : Set
∅       : World
_+1     : World → World
_↑1     : World → World
Name    : World → Set
_⊆_     : World → World → Set

napaKit : DataKit
napaKit = ⟨ World , Name , _↑1 ⟩

```

Figure 2. Core types of NAPA

```

zeroN   : ∀ {α}      → Name (α ↑1)
addN    : ∀ {α} k    → Name α
                                     → Name (α +W k)
subtractN : ∀ {α} k  → Name (α +W k)
                                     → Name α
cmpN     : ∀ {α} ℓ    → Name (α ↑ ℓ)
                                     → Name (∅ ↑ ℓ)
                                     ⊔ Name (α +W ℓ)
_==N_    : ∀ {α} (x y : Name α) → Bool
coerceN  : ∀ {α β}    → α ⊆ β
                                     → Name α → Name β
¬Name∅    : ¬(Name ∅)

syntax addN      k x = x +N k
syntax subtractN k x = x -N k
syntax cmpN      ℓ x = x <N ℓ

```

Figure 3. Core operations on names

2.3 NAPA types

We call our approach NAPA as in **Nameless, Painless**. At the level of types, NAPA exposes the interface shown in figure 2. This is a superset of the `DataKit`, featuring an empty world, a `_+1` operation on worlds described later on, and a notion of world inclusion.

If we look closely at the notion of worlds in the `Fin` approach, we see that it gives us too much information. Indeed a function could look at the given world (by pattern-matching or induction) and behave differently for some values of the world.

On the other hand, `Fin` worlds are too coarse: a single number $ℓ$ is used to represent a set, namely the complete interval $[0..ℓ-1]$.

In NAPA, a world denotes a subset of \mathbb{N} but is kept abstract (no pattern-matching nor induction). Only the empty world \emptyset and two operations are given, namely `_+1` and `_↑1`. Semantically $\alpha +1$ is defined by $\{x +1 \mid x \in \alpha\}$ and $\alpha \uparrow 1$ by $\{0\} \cup (\alpha +1)$.

Internally we represent worlds by lists of Boolean values. Having `true` at the n^{th} position of the list means that n is in the world. Here are the AGDA definitions:

```

World : Set
World = List Bool

∅ : World
∅ = []

_+1 : World → World
α +1 = false :: α

_↑1 : World → World
α ↑1 = true :: α

```

These one step definitions `_+1` and `_↑1` are extended to any number to produce `_+W_` and `_↑W_` of type `World → ℕ → World`.

To the best of our knowledge, making the distinction between two forms of shifting operations for worlds has never been investigated in the context of representing names and binders.

In NAPA the type for names is a refinement of \mathbb{N} , kept abstract by not exporting its definition. From a set point of view, inhabitants of `Name α` are members of α . However we want worlds to form static information only and to be able to erase them before running the programs. To do so we will use a pair of a natural number and a proof it belongs to its world. Here is the membership predicate and the record definition for names:

```

_∈_ : ℕ → World → Set
_ ∈ [] = ⊥
zero ∈ (false :: _) = ⊥
zero ∈ (true :: _) = ⊤
suc n ∈ (_ :: xs) = n ∈ xs

record Name α : Set where
  constructor _,_
  field
    name      : ℕ
    name∈α    : name ∈ α

```

Thanks to our definition of world membership which is canonical, equality of the name field implies equality on whole `Name` values. This means we get provable proof-irrelevance for the `Name` type without requiring an additional axiom.

2.4 Operations on names

The core operations on names are given in figure 3. The simplest name is `zeroN`, it represents 0 and lives in any world shifted by one. One can add any constant to a name in any world with `addN`, the resulting world clearly shows that this function (if parametric in α) does exactly its job. One can do the opposite operation with `subtractN`. Thanks to its precise type this function is total and the inverse of `addN`. Figure 4 depict the core operations we have on names. Starting from the bottom, worlds of the form $\emptyset \uparrow k$ are names which are definitely bound. There value is completely known as stated by arrows with the `Fin k` type. Above we have names that may be bound or free (`Name (α ↑ k)`), a dynamic test (`cmpN`) can tell whether it is bound or not. Above we have names that are known to be greater than k (`Name (α +W k)`), apart from that they are free names. Above we have free names (`Name α`). On the top we have impossible names since they are said to belong to an empty world. From them we can derive everything.

Given any world α , a name in the world $\alpha \uparrow \ell$ is either strictly lower than ℓ (and so also lives in $\emptyset \uparrow \ell$), or greater or equal to ℓ (thus also lives in $\alpha +^W \ell$). This is exactly what the `cmpN` function is about. Given a name, it returns a disjoint sum of names which can be read in two parts. It first gives which side of the disjoint sum it stands, and second it gives a refined version of the input name.

The `coerceN` function will be described in length in the next sub-section. The only way left to extract information from a name in an arbitrary world is to compare it with another name for equality




Figure 4. Operations on names

with $_ ==^N _$. Finally $\neg\text{Name}\emptyset$ asserts that no name lives in the empty world. In a total language this is of great use to tackle impossible cases.

These primitives are enough to show an isomorphism between $\text{Fin } n$ and $\text{Name } (\emptyset \uparrow n)$. From this, every program involving Fin can be translated into our system. This means that our system does not restrict the programmer more than the Fin approach. However as soon as one uses finer types than $\text{Name } (\emptyset \uparrow n)$, then fewer “wrong programs” type-check and more properties hold as we will see later.

2.5 A type for world inclusion witnesses

Having finer types also implies being able to give more types to the same value. Then cheaply moving from different types is a must. We call a function which has an output type different from the input type and behaves as the identity a coercion. To capture a great deal of coercions from a world α to a world β we focus on the case where α is included in β . In NAPA a type $_ \subseteq _$ is introduced to witness the inclusion between two worlds. Then the coerce^N function turns a $\alpha \subseteq \beta$ into an identity function of type $\text{Name } \alpha \rightarrow \text{Name } \beta$. Here is the coerce^N function and its alias $_ \langle - \text{because } _ \rangle$ which is useful to keep the code separated from the typing/proof, in particular what is between the angle brackets can be safely skipped.

$$\begin{array}{l}
\text{coerce}^N : \forall \{ \alpha \beta \} \rightarrow \alpha \subseteq \beta \rightarrow \text{Name } \alpha \rightarrow \text{Name } \beta \\
\text{coerce}^N \alpha \subseteq \beta (x, x \in \alpha) = (x, _ \text{-sound } \alpha \subseteq \beta \ x \ x \in \alpha)
\end{array}$$

$$\begin{array}{l}
\text{infix } 0 _ \langle - \text{because } _ \rangle \\
_ \langle - \text{because } _ \rangle : \forall \{ \alpha \beta \} \rightarrow \text{Name } \alpha \rightarrow \alpha \subseteq \beta \rightarrow \text{Name } \beta \\
_ \langle - \text{because } _ \rangle n \text{ pf} = \text{coerce}^N \text{ pf } n
\end{array}$$

The inclusion relation also expresses the emptiness of a world α by using $\alpha \subseteq \emptyset$. We use this definition of emptiness as opposed to an equality with the empty world for two reasons. First \emptyset is not the only empty world ($\emptyset + 1$ is empty as well). Second having world equalities would require new definitions and would be heavy in comparison to the use of inclusions. Combining coerce^N and $\neg\text{Name}\emptyset$ turns any contradiction on names into an inclusion problem, reusing any automation done on this side.

$$\begin{array}{l}
\neg\text{Name} : \forall \{ \alpha \} \rightarrow \alpha \subseteq \emptyset \rightarrow \neg(\text{Name } \alpha) \\
\neg\text{Name } \alpha \subseteq \emptyset = \neg\text{Name}\emptyset \circ \text{coerce}^N \alpha \subseteq \emptyset
\end{array}$$

$$\begin{array}{l}
\subseteq \text{-refl} : \forall \{ \alpha \} \rightarrow \alpha \subseteq \alpha \\
\subseteq \text{-trans} : \forall \{ \alpha \beta \gamma \} \rightarrow \alpha \subseteq \beta \rightarrow \beta \subseteq \gamma \rightarrow \alpha \subseteq \gamma \\
\subseteq \text{-}\emptyset : \forall \{ \alpha \} \rightarrow \emptyset \subseteq \alpha \\
\subseteq \text{-}\emptyset + 1 : \emptyset + 1 \subseteq \emptyset \\
\subseteq \text{-}\uparrow 1 \text{-}\uparrow 1 : \forall \{ \alpha \beta \} \rightarrow \alpha \subseteq \beta \leftrightarrow \alpha \uparrow 1 \subseteq \beta \uparrow 1 \\
\subseteq \text{-}\uparrow 1 \text{-}\uparrow 1 : \forall \{ \alpha \beta \} \rightarrow \alpha \subseteq \beta \leftrightarrow \alpha + 1 \subseteq \beta + 1 \\
\subseteq \text{-}\uparrow 1 \text{-}\uparrow 1 : \forall \{ \alpha \} \rightarrow \alpha + 1 \subseteq \alpha \uparrow 1
\end{array}$$

Figure 5. Rules for world inclusion

We expose the inclusion relation with a set of rules given in figure 5. This states that the inclusion relation is reflexive, transitive, and has the empty world as lowest element. The $\subseteq \text{-}\emptyset + 1$ rule states that $\emptyset + 1$ is empty. The inclusion is preserved both ways by the contexts $_ \uparrow 1$ and $_ + 1$. Finally $_ + 1$ can be weakened in $_ \uparrow 1$. This accounts for the fact that $\alpha \uparrow 1$ means $\{0\} \cup (\alpha + 1)$ and so is a superset of $\alpha + 1$. This set of rules has been shown sound according to a semantic definition of inclusion, namely $\forall x \rightarrow x \in \alpha \rightarrow x \in \beta$. On top of these base rules, we derived others that we omit to define here but some are used in the code.

Sometimes one has to build complex inclusion witnesses. While inference would be of great effect here, we propose a modest syntactic tool to build them, namely the $\subseteq \text{-Reasoning}$ module. It gives access to the transitivity $\subseteq \text{-trans}$ in a style which focuses more on the intermediate states of the reasoning rather than the steps. The syntax is a list of worlds interleaved by inclusion witnesses, with two \blacksquare around the last world. There is one example of its use in the following section. The code for $\subseteq \text{-Reasoning}$ is given below for reference and can be safely skipped.

```

module  $\subseteq \text{-Reasoning}$  where
  infix 2 finally
  infixr 2  $\_ \subseteq \langle \_ \rangle \_$ 

   $\_ \subseteq \langle \_ \rangle \_ : \forall x \{y z\} \rightarrow x \subseteq y \rightarrow y \subseteq z \rightarrow x \subseteq z$ 
   $\_ \subseteq \langle x \subseteq y \rangle y \subseteq z = \subseteq \text{-trans } x \subseteq y \ y \subseteq z$ 

  finally :  $\forall x \ y \rightarrow x \subseteq y \rightarrow x \subseteq y$ 
  finally  $\_ \_ x \subseteq y = x \subseteq y$ 

  syntax finally x y  $x \subseteq y = x \subseteq \langle x \subseteq y \rangle \blacksquare y \blacksquare$ 

```

2.6 Singleton worlds!

We said that our worlds denote finite subsets of \mathbb{N} and are more precise than in the Fin approach. Actually they can be as precise as we wish, since any subset of \mathbb{N} can be described by our operations on worlds (\emptyset , $_ + 1$, and $_ \uparrow 1$). In particular they can be singleton worlds. From singleton worlds we build singleton types for names:

World^s : $\mathbb{N} \rightarrow \text{World}$
World^s n = $\emptyset \uparrow 1 +^W n$

Name^s : $\mathbb{N} \rightarrow \text{Set}$
Name^s = Name \circ World^s

$_{}^s$: $\forall n \rightarrow \text{Name}^s n$
 $_{}^s n = \text{zero}^N +^N n$

Singleton worlds not only exist, they are also preserved by our two updating operations, namely add^N and subtract^N .

add^s : $\forall \{n\} k \rightarrow \text{Name}^s n \rightarrow \text{Name}^s (k + \mathbb{N} n)$
 $\text{add}^s \{n\} k x = \text{add}^N k x$
 $\langle \text{-because } \subseteq \text{-assoc-+ } \subseteq \text{-refl } n k \text{-} \rangle$

subtract^s : $\forall \{n\} k \rightarrow \text{Name}^s (k + \mathbb{N} n) \rightarrow \text{Name}^s n$
 $\text{subtract}^s \{n\} k x = \text{subtract}^N k x$
 $\langle \text{-because } \subseteq \text{-assoc-+}' \subseteq \text{-refl } n k \text{-} \rangle$

3. Using NAPA: examples and advanced operations

3.1 Some convenience functions

Here are a few functions built on top of the interface (without using the concrete representation of names). suc^N is $\text{add}^N 1$ and $\text{suc}^N \uparrow$ is a variant that includes a coercion from $\alpha + 1$ to $\alpha \uparrow 1$. The function $_{}^N$ turns a number n into a name that inhabits any world with at least $n+1$ consecutive binders.

suc^N : $\forall \{\alpha\} \rightarrow \text{Name } \alpha \rightarrow \text{Name } (\alpha + 1)$
 $\text{suc}^N = \text{add}^N 1$

$\text{suc}^N \uparrow$: $\forall \{\alpha\} \rightarrow \text{Name } \alpha \rightarrow \text{Name } (\alpha \uparrow 1)$
 $\text{suc}^N \uparrow = \text{coerce}^N \subseteq \text{-+1}\uparrow 1 \circ \text{suc}^N$

$_{}^N$: $\forall \{\alpha\} n \rightarrow \text{Name } (\alpha \uparrow \text{suc } n)$
 $_{}^N \{\alpha\} n = \text{zero}^N +^N n$
 $\langle \text{-because } \alpha \uparrow 1 +^N n \subseteq \langle \subseteq \text{-+}\uparrow n \rangle$
 $\alpha \uparrow 1 \uparrow n \subseteq \langle \subseteq \text{-exch-}\uparrow\uparrow 1 n \rangle \blacksquare$
 $\alpha \uparrow \text{suc } n \blacksquare \text{-} \rangle$

where open $\subseteq \text{-Reasoning}$

We call ℓ -bound, a name bound somewhere in the scope of ℓ binders. We call ℓ -free, a name free for all ℓ binders. In other words, a de Bruijn index is ℓ -bound if it is strictly less than ℓ ; it is ℓ -free otherwise.

The exportName function tells whether a given name is ℓ -bound or ℓ -free. In case the name is free, an exported version of it is returned. This function forms the base case of exporting functions like exportTm explained later on.

-- Partial functions from A to B
A $\rightarrow?$ B = A \rightarrow Maybe B

exportName : $\forall \{\alpha\} \ell \rightarrow \text{Name } (\alpha \uparrow \ell)$
 $\rightarrow?$ Name α

$\text{exportName } \ell x$
with $x \ll^N \ell$
... | $\text{inj}_1 _{} = \text{nothing}$
... | $\text{inj}_2 x' = \text{just } (x' _{}^N \ell)$

The function $\text{shiftName } \ell k \text{ pf}$ shifts its argument by k if this name is ℓ -free, otherwise it leaves the ℓ -bound name untouched. This function makes use of cmp^N and coerces the outputs to the required type. It also perform a coercion on the fly, giving extra flexibility for free.

shiftName : $\forall \{\alpha\} \ell k \rightarrow (\alpha +^W k) \subseteq \beta$
 $\rightarrow \text{Name } (\alpha \uparrow \ell)$
 $\rightarrow \text{Name } (\beta \uparrow \ell)$

$\text{shiftName } \ell k \text{ pf } x$
with $x \ll^N \ell$
... | $\text{inj}_1 x' = x'$
 $\langle \text{-because } \text{pf}_1 \text{-} \rangle$
... | $\text{inj}_2 x' = x' +^N k$
 $\langle \text{-because } \text{pf}_2 \text{-} \rangle$
where
 $\text{pf}_1 = \subseteq \text{-cong-}\uparrow \subseteq \emptyset \ell$
 $\text{pf}_2 = \subseteq \text{-trans } (\subseteq \text{-exch-+}\uparrow \subseteq \text{-refl } \ell k)$
 $(\subseteq \text{-ctx-}\uparrow \text{pf } \ell)$

The $\text{protect}\uparrow$ function shifts a name transformer. Let f be a function from names to names. The function $\text{protect}\uparrow \ell f$ is a version of f that is applicable under ℓ binders. Let x be a name under ℓ binders. When x is ℓ -bound, it is left untouched by $\text{protect}\uparrow$. When x is ℓ -free, we can subtract ℓ to x , give it to f , and then add ℓ to get the result. By combining $\text{protect}\uparrow$ and add^N one obtain an alternative implementation of shiftName called $\text{shiftName}'$. However shiftName is more efficient since it avoids to subtracting ℓ to add it back after adding k .

$\text{protect}\uparrow$: $\forall \{\alpha \beta\} \ell$
 $\rightarrow (\text{Name } \alpha \rightarrow \text{Name } \beta)$
 $\rightarrow (\text{Name } (\alpha \uparrow \ell) \rightarrow \text{Name } (\beta \uparrow \ell))$

$\text{protect}\uparrow \ell f x$
with $x \ll^N \ell$
... | $\text{inj}_1 x' = x'$
 $\langle \text{-because } \subseteq \text{-cong-}\uparrow \subseteq \emptyset \ell \text{-} \rangle$
... | $\text{inj}_2 x' = f (x' _{}^N \ell) +^N \ell$
 $\langle \text{-because } \subseteq \text{-+}\uparrow \ell \text{-} \rangle$

$\text{shiftName}'$: $\forall \{\alpha \beta\} \ell k \rightarrow (\alpha +^W k) \subseteq \beta$
 $\rightarrow \text{Name } (\alpha \uparrow \ell) \rightarrow \text{Name } (\beta \uparrow \ell)$
 $\text{shiftName}' \ell k \text{ pf} = \text{protect}\uparrow (\text{coerce}^N \text{pf} \circ \text{add}^N k) \ell$

3.2 Building terms

Building terms in NAPA is as easy as building them in the other nameless approaches we have seen. The structure is exactly the same, and the variables are made of numbers (of type \mathbb{N}) using $_{}^N$. Below we define the representation of the identity function as idTm , the application operator as appTm and finally the composition function as compTm :

idTm : $\forall \{\alpha\} \rightarrow \text{Tm } \alpha$
 $\text{idTm} = \lambda (V (0^N))$

appTm : $\forall \{\alpha\} \rightarrow \text{Tm } \alpha$
 $\text{appTm} = \lambda (\lambda (V (1^N)) \cdot V (0^N))$

compTm : $\forall \{\alpha\} \rightarrow \text{Tm } \alpha$
 $\text{compTm} = \lambda (\lambda (\lambda (V (2^N)) \cdot (V (1^N) \cdot V (0^N))))$

3.3 Computing free variables

Our first example of functions over terms simply computes the list of free variables in the input term. The `fv` function below, while straightforward, has the subtle cases of binding constructs (λ and `Let`). In these cases we have to remove the bound variable from the list of free variables given by the recursive call. In this nameless representation this amounts to removing occurrences of 0 and subtract 1 to other name occurrences. This is done by the function `rm0` which calls `{agda|exportName 1}` on each element of the list and merges the results. Note that forgetting to remove the bound variable will result in a type error. In the same vein the typing of `fv` ensures that all returned variables do appear free in the given term.

```

rm0 : ∀ {α} → List (Name (α ↑ 1))
      → List (Name α)
rm0 [] = []
rm0 (x :: xs) with exportName 1 x
...       | just x' = x' :: rm0 xs
...       | nothing = rm0 xs

fv : ∀ {α} → Tm α → List (Name α)
fv (V x)      = [ x ]
fv (fct · arg) = fv fct ++ fv arg
fv (λ t)      = rm0 (fv t)
fv (Let t u)  = fv t ++ rm0 (fv u)

```

3.4 Generic traversal

In order to build multiple traversal functions at once, we first define a generic function based on ideas from [8]. It is essentially a `map/subst` function where the free variables are transformed by a user-supplied function. Moreover a class of effects (an applicative functor) is allowed during the traversal. An applicative functor is halfway between a functor and a monad. Like a monad, an applicative functor has a unit called `pure`. It allows to embed any pure value as a potentially effectful one. The second operation called `_@_` is an effectful application, taking both an effectful function and argument and resulting in an effectful result.

```

module TraverseTm
  {E} (E-app : Applicative E)
  {α β} (trName : ∀ ℓ → Name (α ↑ ℓ)
                 → E (Tm (β ↑ ℓ)))

  where
  open Applicative E-app

  tr : ∀ ℓ → Tm (α ↑ ℓ) → E (Tm (β ↑ ℓ))
  tr ℓ (V x)      = trName ℓ x
  tr ℓ (t · u)    = pure _ · _ @ tr ℓ t @ tr ℓ u
  tr ℓ (λ t)      = pure λ @ tr (suc ℓ) t
  tr ℓ (Let t u)  = pure Let @ tr ℓ t
                  @ tr (suc ℓ) u

  trTm : Tm α → E (Tm β)
  trTm = tr 0

```

To put this traversal at work we successively instantiate some of its arguments. For instance you may have noticed the special case we made for variables. Mapping names to terms will bring us capture avoiding substitution almost for free. However, in the mean time we build `trTm'` which maps names to names. It does so by applying `pure V` to the name to name function:

```

open TraverseTm

trTm' :
  ∀ {E} (E-app : Applicative E) {α β}
  (trName : ∀ ℓ → Name (α ↑ ℓ)
            → E (Name (β ↑ ℓ)))
  → Tm α → E (Tm β)
trTm' E-app trName
  = trTm E-app (λ ℓ x → pure V @ trName ℓ x)
  where open Applicative E-app

```

Renaming functions In many functions over terms the handling of variables shares a common part. Given a variable under ℓ binders, we test if the variable is ℓ -bound. If so we leave it untouched, otherwise we subtract ℓ and go on a specific processing after which we add ℓ again to the free variables of the result. The traversal function is augmented by this processing of bound variables to build `renameTmA`. Then by picking either the identity functor (`id-app`), or the `Maybe` one we build two functions (a total and a partial one) to perform renamings, namely `renameTm` and `renameTm?`:

```

renameTmA : ∀ {E} (E-app : Applicative E)
            {α β} (θ : Name α → E (Name β))
            → Tm α → E (Tm β)
renameTmA E-app θ
  = trTm' E-app (protect↑A E-app θ)

```

```

renameTm : ∀ {α β} → (Name α → Name β)
           → Tm α → Tm β
renameTm θ = trTm' id-app (protect↑ θ)
-- or
-- renameTm = renameTmA id-app

```

```

renameTm? : ∀ {α β} → (Name α →? Name β)
           → Tm α →? Tm β
renameTm? = renameTmA Maybe.applicative

```

Lifting name functions Any operation on names can now be lifted to terms:

```

addTm : ∀ {α} k → Tm α → Tm (α +W k)
addTm = renameTm ∘ addN

subtractTm : ∀ {α} k → Tm (α +W k) → Tm α
subtractTm = renameTm ∘ subtractN

exportTm? : ∀ {α} ℓ → Tm (α ↑ ℓ) →? Tm α
exportTm? = renameTm? ∘ exportName

```

While `coerceN` can be lifted to terms in the same way, using `trTm'` directly enables to by-pass the `protect↑` dynamic tests and directly “protect” the inclusion witness with an appropriate inclusion rule.

```

coerceTm : ∀ {α β} → α ⊆ β → Tm α → Tm β
coerceTm pf = trTm' id-app (coerceN ∘ ⊆-cong-↑ pf)
-- or less efficiently:
-- coerceTm = renameTm ∘ coerceN

```

Lifting add^N to terms can be done more efficiently (than using $\text{protect}\uparrow$) as well. Here the dynamic test performed by $\text{protect}\uparrow$ is necessary. However when the name is ℓ -free we subtract ℓ to add it back after adding k . The function shiftName avoids this extra computation, hence the following shiftTm :

```

shiftTm :  $\forall \{\alpha \beta\} k \rightarrow (\alpha +^N k) \subseteq \beta \rightarrow \text{Tm } \alpha \rightarrow \text{Tm } \beta$ 
shiftTm k p = trTm' id-app ( $\lambda \ell \rightarrow \text{shiftName } \ell k p$ )
-- or less efficiently:
-- shiftTm k pf = renameTm (coerceN pf  $\circ$  addN k)

```

A special case of renameTm ? is the so-called closeTm ?. This function takes a term in any world and checks if the term is closed. If so, the same term is returned in the empty world. Otherwise the function fails by returning nothing :

```

closeTm? :  $\forall \{\alpha\} \rightarrow \text{Tm } \alpha \rightarrow? \text{Tm } \emptyset$ 
closeTm? = renameTm? (const nothing)

```

Capture avoiding substitution To implement capture avoiding substitution for the Tm type we solely need a specific trName function for trTm . Here substitutions are represented as functions from names to terms. The function substVarTm handles the case for variables. This function is very close to $\text{protect}\uparrow$ but extended to functions returning terms.

```

substVarTm :  $\forall \{\alpha \beta\} \rightarrow (\text{Name } \alpha \rightarrow \text{Tm } \beta) \rightarrow$ 
              $\forall \ell \rightarrow \text{Name } (\alpha \uparrow \ell) \rightarrow \text{Tm } (\beta \uparrow \ell)$ 
substVarTm f  $\ell x$ 
  with  $x <^N \ell$ 
... | inj1 x' = V (x'
  (-because  $\subseteq$ -cong- $\uparrow \subseteq$ - $\emptyset \ell$  -))
... | inj2 x' = shiftTm  $\ell$  ( $\subseteq$ - $\uparrow$   $\ell$ ) (f (x'  $\uparrow^N \ell$ ))

```

The main function substTm instantiates trTm with the identity applicative functor and substVarTm :

```

substTm :  $\forall \{\alpha \beta\} \rightarrow (\text{Name } \alpha \rightarrow \text{Tm } \beta)$ 
          $\rightarrow \text{Tm } \alpha \rightarrow \text{Tm } \beta$ 
substTm = trTm id-app  $\circ$  substVarTm

```

As an illustration, the function β -red takes the body of a λ -abstraction and calls substTm with the substitution that replaces 0 with a received replacement term:

```

 $\beta$ -red :  $\forall \{\alpha\} \rightarrow \text{Tm } (\alpha \uparrow 1) \rightarrow \text{Tm } \alpha \rightarrow \text{Tm } \alpha$ 
 $\beta$ -red  $\{\alpha\} f a = \text{substTm } (\phi \circ \text{exportName } 1) f$ 
  where  $\phi : \text{Maybe } (\text{Name } \alpha) \rightarrow \text{Tm } \alpha$ 
          $\phi$  (just x) = V x
          $\phi$  nothing = a

```

3.5 Deciding term α -equivalence

Our next example focuses on comparison of terms. We first define αEq , where $\alpha\text{Eq } F$ is the type of functions comparing F structures.

```

 $\alpha\text{Eq} : (F : \text{World} \rightarrow \text{Set}) (\alpha \beta : \text{World}) \rightarrow \text{Set}$ 
 $\alpha\text{Eq } F \alpha \beta = F \alpha \rightarrow F \beta \rightarrow \text{Bool}$ 

```

We show that all the subtle work is done at the level of names in a separate and reusable function called αeqName . This function takes a function that compares two free names and builds one that compares two names under ℓ bindings. It does so by comparing both of them to ℓ . If they are both bound they can be safely compared using $_ =^N _$ since they now are of the same type. If they

are both free, they can be compared using the function received as argument. Otherwise they are different.

```

 $\alpha\text{eqName} : \forall \{\alpha \beta\} \ell \rightarrow \alpha\text{Eq } \text{Name } \alpha \beta$ 
              $\rightarrow \alpha\text{Eq } \text{Name } (\alpha \uparrow \ell) (\beta \uparrow \ell)$ 
 $\alpha\text{eqName } \ell \Gamma x y$  with  $x <^N \ell \mid y <^N \ell$ 
... | inj1 x' | inj1 y' = x'  $\stackrel{N}{=}$  y'
... | inj2 x' | inj2 y' =  $\Gamma$  (x'  $\uparrow^N \ell$ ) (y'  $\uparrow^N \ell$ )
... | _ | _ = false

```

The αeqTm function structurally compares two terms in a simple way, only keeping track of the number of traversed binders and calling αeqName at variables.

```

 $\alpha\text{eqTm} : \forall \{\alpha \beta\} \rightarrow \alpha\text{Eq } \text{Name } \alpha \beta \rightarrow \alpha\text{Eq } \text{Tm } \alpha \beta$ 
 $\alpha\text{eqTm } \{\alpha\} \{\beta\} \Gamma = \text{go } 0$  where
  go :  $\forall \ell \rightarrow \alpha\text{Eq } \text{Tm } (\alpha \uparrow \ell) (\beta \uparrow \ell)$ 
  go  $\ell$  (V x) (V y) =  $\alpha\text{eqName } \ell \Gamma x y$ 
  go  $\ell$  (t · u) (v · w) = go  $\ell$  t v  $\wedge$  go  $\ell$  u w
  go  $\ell$  ( $\lambda$  t) ( $\lambda$  u) = go (suc  $\ell$ ) t u
  go  $\ell$  (Let t u) (Let v w) = go  $\ell$  t v
                                $\wedge$  go (suc  $\ell$ ) u w
  go _ _ _ = false

```

4. Soundness using logical relations

Since our library is written in a type-safe language, one may wonder what soundness properties are to be proved. Moreover our names are indexed by worlds and hold membership proofs. The functions over names are given precise types and have been shown to be type-safe. However we still wish to show that our library respects a model of good behavior with respect to names and binders. Our model is based on logical relations indexed/directed by types. This technique [4] is independent of this work and enables to define a notion of program equivalence. We use this technique to capture good behavior of functions involving names and binders. Using this technique, the set of specific definitions is kept to a minimum of one per introduced type (World , Name , and $_ \subseteq _$). One proof per value introduced has to be done, which keeps the development modular and forward-compatible to the addition of more features.

This section is organized as follows. First we recall the basics of logical relations. Then we give a toy example to practice a bit. Then relations for worlds, names, and world inclusions are given. Finally we make use of the construction to obtain free theorems [12] about world-polymorphic functions over the Tm type.

4.1 Recap of the framework

A relation is said to be type-directed when it is recursively defined on the structure of types. Let \mathcal{R} be such a type-directed relation, and let τ be a type. Then, $\mathcal{R}\tau$ is a relation on values of type τ , namely $\mathcal{R}\tau : \tau \rightarrow \tau \rightarrow \text{Set}$. Recall that Set is also the type of propositions in AGDA.

A type-directed relation is called “logical” when the case for functions is defined extensionally. Here extensionally means that two functions are related when they produce related results out of related arguments. Let A_r be a relation for the arguments and B_r a relation for results. Two functions f_1 and f_2 are related if and only if for every pair of arguments (x_1, x_2) related by A_r , the results $f_1 x_1$ and $f_2 x_2$ are related by B_r . This definition can be given in AGDA as well:

```

RelatedFunctions Ar Br f1 f2 =
   $\forall \{x_1 x_2\} \rightarrow A_r x_1 x_2$ 
   $\rightarrow B_r (f_1 x_1) (f_2 x_2)$ 

```


We say that a program or a value fits a logical relation when it is related to itself by the relation indexed by its type. We say that a logical relation is universal if every well-typed program fits the logical relation. This notion of universality was originally coined by John Reynolds as the “Abstraction Theorem” [11]. We call the “AGDA logical relation” the one defined by Bernardy et al. [4] for a \mathcal{PTS} (pure type system) and naturally extended as they suggest to other features of AGDA. While no complete mechanized proof has been done for this we will consider the AGDA logical relation as universal.

To simplify matters, the definitions shown here are not universe polymorphic. You can find universe-polymorphic definitions in our complete AGDA development [9].

To formally define a logical relation indexed by types, a common technique is to first inductively define the structure of types. This is known as a “universe of codes” \mathcal{U} . Then one defines a function called E1 from codes to types. Finally one defines by induction a function called $\llbracket _ \rrbracket$ from codes to relations on elements of type described by the given code. In AGDA the $\llbracket _ \rrbracket$ function has the following type: $(\tau : \mathcal{U}) \rightarrow \text{E1 } \tau \rightarrow \text{E1 } \tau \rightarrow \text{Set}$. Because, when types contain variables, a good deal of complexity is added to this scheme, we opt for a lighter scheme. We do not define \mathcal{U} , E1 , and $\llbracket _ \rrbracket$.

Instead, for each type constructor κ , we define a relation $\llbracket \kappa \rrbracket$. For the function type constructor the `RelatedFunctions` definition above is a good start. Actually this is a fine definition for non-dependent functions. The dependent version of `RelatedFunctions`, called $\llbracket \Pi \rrbracket$ here, passes the relation argument A_r x_1 x_2 called x_r , to the relation for results B_r . In short the relation for results now depends on the relation for arguments. Here is the definition in AGDA:

$$\llbracket \Pi \rrbracket A_r B_r f_1 f_2 = \forall \{x_1 x_2\} (x_r : A_r x_1 x_2) \rightarrow B_r x_r (f_1 x_1) (f_2 x_2)$$

Note that this definition generalizes the case of non-dependent functions and universal quantifications as well. For non-dependent functions we simply provide a relation for results which ignores its first argument (equivalent to `RelatedFunctions` and noted $\llbracket _ \rrbracket$ from now on). For universal quantifications, since the arguments are types, all we need is a relation for types (members of Set_0) themselves. Following our convention we call this relation $\llbracket \text{Set}_0 \rrbracket$, its definition is the set of relations between its arguments:

$$\llbracket \text{Set}_0 \rrbracket : \text{Set}_0 \rightarrow \text{Set}_0 \rightarrow \text{Set}_1 \\ \llbracket \text{Set}_0 \rrbracket A_1 A_2 = A_1 \rightarrow A_2 \rightarrow \text{Set}_0$$

For reference, full definitions for core type theory are given in figure 6. We now have all the building blocks of the meta-function $\llbracket _ \rrbracket$ and we will not materialize it more here. We will apply the $\llbracket _ \rrbracket$ function manually. To this end it suffices to replace each constructor κ by $\llbracket \kappa \rrbracket$, each non dependent arrow by $\llbracket _ \rrbracket$, each dependent arrow $(x : A) \rightarrow B$ by $\langle x_r : \llbracket A \rrbracket \rangle \llbracket _ \rrbracket \llbracket B \rrbracket$. By convention we subscript the variables by r . Applications are translated to applications. Because of dependent types, this translation has to be extended to all terms but we will not do it here. Finally here are a few examples of the manual use of the $\llbracket _ \rrbracket$ function:

$$\llbracket \mathbb{N} \rightarrow \mathbb{N} \rightarrow \text{Bool} \rrbracket = \\ \llbracket \mathbb{N} \rrbracket \llbracket _ \rrbracket \llbracket \mathbb{N} \rrbracket \llbracket _ \rrbracket \llbracket \text{Bool} \rrbracket = \\ \lambda f_1 f_2 \rightarrow \\ \forall \{x_1 x_2\} (x_r : \llbracket \mathbb{N} \rrbracket x_1 x_2) \\ \{y_1 y_2\} (y_r : \llbracket \mathbb{N} \rrbracket y_1 y_2) \\ \rightarrow \llbracket \text{Bool} \rrbracket (f_1 x_1 y_1) (f_2 x_2 y_2)$$

```
data  $\llbracket \perp \rrbracket$  :  $\llbracket \text{Set}_0 \rrbracket \perp \perp$  -- no constructors

data  $\llbracket \text{Bool} \rrbracket$  :  $\llbracket \text{Set}_0 \rrbracket \text{Bool Bool}$  where
   $\llbracket \text{true} \rrbracket$  :  $\llbracket \text{Bool} \rrbracket \text{true true}$ 
   $\llbracket \text{false} \rrbracket$  :  $\llbracket \text{Bool} \rrbracket \text{false false}$ 

data  $\llbracket \mathbb{N} \rrbracket$  :  $\llbracket \text{Set}_0 \rrbracket \mathbb{N} \mathbb{N}$  where
   $\llbracket \text{zero} \rrbracket$  :  $\llbracket \mathbb{N} \rrbracket \text{zero zero}$ 
   $\llbracket \text{suc} \rrbracket$  :  $(\llbracket \mathbb{N} \rrbracket \llbracket \_ \rrbracket \llbracket \mathbb{N} \rrbracket) \text{suc suc}$ 

data  $\llbracket \_ \uplus \rrbracket$   $\{A_1 A_2 B_1 B_2\}$   $(A_r : \llbracket \text{Set}_0 \rrbracket A_1 A_2)$ 
   $(B_r : \llbracket \text{Set}_0 \rrbracket B_1 B_2)$  :
   $A_1 \uplus B_1 \rightarrow A_2 \uplus B_2 \rightarrow \text{Set}_0$  where
   $\llbracket \text{inj}_1 \rrbracket$  :  $(A_r \rightarrow A_r \llbracket \_ \uplus \rrbracket B_r) \text{inj}_1 \text{inj}_1$ 
   $\llbracket \text{inj}_2 \rrbracket$  :  $(B_r \rightarrow A_r \llbracket \_ \uplus \rrbracket B_r) \text{inj}_2 \text{inj}_2$ 
```

Figure 7. Logical relations for data types

```
 $\llbracket (A : \text{Set}_0) \rightarrow A \rightarrow A \rrbracket =$ 
 $\llbracket \Pi \rrbracket \llbracket \text{Set}_0 \rrbracket (\lambda A_r \rightarrow A_r \llbracket \_ \rrbracket A_r) =$ 
 $\lambda f_1 f_2 \rightarrow$ 
 $\forall \{A_1 A_2\} (A_r : A_1 \rightarrow A_2 \rightarrow \text{Set}_0)$ 
 $\{x_1 x_2\} (x_r : A_r x_1 x_2)$ 
 $\rightarrow A_r (f_1 A_1 x_1) (f_2 A_2 x_2)$ 

-- Using the notation instead of  $\llbracket \Pi \rrbracket$ :
 $\llbracket (A : \text{Set}_0) \rightarrow \text{List } A \rrbracket =$ 
 $\langle A_r : \llbracket \text{Set}_0 \rrbracket \rangle \llbracket \_ \rrbracket \llbracket \text{List} \rrbracket A_r =$ 
 $\lambda l_1 l_2 \rightarrow$ 
 $\forall \{A_1 A_2\} (A_r : A_1 \rightarrow A_2 \rightarrow \text{Set}_0)$ 
 $\rightarrow \llbracket \text{List} \rrbracket A_r (l_1 A_1) (l_2 A_2)$ 
```

We now have the definition of the AGDA logical relation for the core type theory part. It extends nicely to inductive data types and records. The process is as follows: for each constructor κ of type τ , declare a new constructor $\llbracket \kappa \rrbracket$ whose type is $\llbracket \tau \rrbracket \kappa$. This process applies to type constructors and data constructors of data types, and type constructors and fields of record types. For reference, the logical relations for data types we use in this development are in figure 7.

4.2 An example: booleans represented by numbers

We wish to explain how logical relations can help build a safe interface to an abstract type. To do so we introduce a tiny example about booleans represented using natural numbers. We want 0 to represent `false` and any other number to represent `true`. Therefore the boolean disjunction can be implemented using addition. We show that logical relations help build a model, ensure that a given implementation respects this model, and finally show that a client that uses only the interface will also respect the model.

Note however that this is a toy example in several ways. There are no polymorphic functions in the interface, so no interesting free theorems are to be expected. While we could prove the safety by defining a representation predicate in unary style, the logical relations approach is different. It relies on comparing concrete data as opposed to mapping to abstract data. The unary construction would allow for a simpler construction, however this oversimplifies the problem here and would be no longer useful for proving NAPA.

Our tiny implementation of booleans using natural numbers is given below. It contains a type `B` that we want to keep abstract. It contains obvious definitions for `true`, `false`, and the disjunction

```

[[Set0]] : ∀ (A1 A2 : Set0) → Set1
[[Set0]] A1 A2 = A1 → A2 → Set0

[[Set1]] : ∀ (A1 A2 : Set1) → Set2
[[Set1]] A1 A2 = A1 → A2 → Set1

_[->]_ : ∀ {A1 A2 B1 B2} → [[Set0]] A1 A2 → [[Set0]] B1 B2 → [[Set0]] (A1 → B1) (A2 → B2)
Ar [->] Br = λ f1 f2 → ∀ {x1 x2} → Ar x1 x2 → Br (f1 x1) (f2 x2)

infixr 0 _[->]_

[[II]] : ∀ {A1 A2} (Ar : [[Set0]] A1 A2)
        {B1 B2} (Br : (Ar [->] [[Set0]]) B1 B2)
        → ((x : A1) → B1 x) → ((x : A2) → B2 x) → Set1
[[II]] Ar Br = λ f1 f2 → ∀ {x1 x2} (xr : Ar x1 x2) → Br xr (f1 x1) (f2 x2)

syntax [[II]] Ar (λ xr → f) = ⟨ xr : Ar ⟩ [->] f

[[V]] : ∀ {A1 A2} (Ar : [[Set0]] A1 A2)
        {B1 B2} (Br : ([[Set0]] [->] [[Set0]]) B1 B2)
        → [[Set1]] ({x : A1} → B1 x) ({x : A2} → B2 x)
[[V]] Ar Br = λ f1 f2 → ∀ {x1 x2} (xr : Ar x1 x2) → Br xr (f1 {x1}) (f2 {x2})

syntax [[V]] Ar (λ xr → f) = ∀⟨ xr : Ar ⟩ [->] f

```

Figure 6. Logical relations for core types

`_∨_`. It intentionally has a dubious function `is42?` which breaks our still informal expectations from such a module.

```

B : Set
B = ℕ

false : B
false = 0

true : B
true = 1

_∨_ : B → B → B
m ∨ n = m + n

is42? : B → B
is42? 42 = true
is42? _ = false

```

The next step is to define our expectations. To do so, we give a binary relation which tells when two `B` values have the same meaning. We do so with an (inductive) data type named `[[B]]` which states that `0` is related only with itself, and that any two non zero numbers are related:

```

data [[B]] : B → B → Set where
  [[false]] : [[B]] 0 0
  [[true]]  : ∀ {m n} → [[B]] (suc m) (suc n)

```

When plugged into the machinery of logical relations this single definition suffices to define a complete model of well-typed programs. However, the plumbing requires some care. While the AGDA logical relation is universal, we have no such guarantee about the AGDA logical relation where the relation for `B` is no

longer `[[N]]` but `[[B]]`. Fortunately changing the relation at a given type (`B` here) can be done safely. All we have to do is to consider programs abstracted away from `B` and its operations: `true`, `false` and `_∨_`. This can be done either through a mechanism for abstract types, or by requiring the client to be a function taking the implementation for `B` and its operations as argument.

However to use `[[B]]` as the relation for `B`, we have to show that the definitions which make use of the representation of `B` actually fit the relation. Since `[[true]]` and `[[false]]` are obvious witnesses for `true` and `false`, only `_∨_` and `is42?` are left to be proved. Each time the goal to prove is mechanical: wrap the type with `[[·]]` on each constructor and put the name of the function twice to state we want it to be related to itself. Here is the definition for `_[[V]]_`:

```

_[[V]]_ : ([[B]] [->] [[B]] [->] [[B]]) _∨_ _∨_

```

The type of `_[[V]]_` means that given inputs related in the model, the results are related in the model as well. Once unfolded the type looks like:

```

_[[V]]_ : ∀ {x1 x2} (xr : [[B]] x1 x2)
        {y1 y2} (yr : [[B]] y1 y2)
        → [[B]] (x1 ∨ y1) (x2 ∨ y2)

```

The fact that input arguments come as implicit arguments will greatly shorten definitions. Now, thanks to the inductive definition of `_+_`, simply pattern-matching on the first relation suffices to reduce the goal, and allows this nice looking definition where we see the usual lazy definition of the left biased disjunction:

```

[[false]] [[V]] x = x
[[true]]  [[V]] _ = [[true]]

```

Let us now consider a proof for the `is42?` function. Fortunately there is no such proof since this function gladly breaks the intended abstraction. Instead we simply prove its negation by exhibiting that

given two related inputs (42 and 27) we get non related outputs (is42? 42 = 1 and is42? 27 = 0).

```

¬ [[is42?]] : ¬ (([[B]] [[→]] [[B]]) is42? is42?)
¬ [[is42?]] [[is42?]] with [[is42?]] {42} {27} [[true]]
...
| () -- absurd

```

Note that is42? is rejected by our model with no considerations about the other exported functions. Indeed with another implementation of `_v_` there would be no way to produce 42 and so no way to expose the wrong behavior of is42? using the interface. Using a model provides a better forward compatibility and enables proofs to be done in a modular way.

4.3 Relations for NAPA

For NAPA, we apply the same process as with booleans. We define our expectations, by defining relations for introduced types (worlds, names, and inclusions). Finally we prove that each value/function exported fits the relation.

4.3.1 Relations for NAPA types

Valid names are those which belong to their worlds, names with the same meaning are those related by the relation between their worlds. What matters is not just the fact that two worlds are related, but “how” they are related. Indeed this will dictate when two names are related. We need to define a set of valid relations between worlds. The more relations are accepted, the more power it gives to the free theorems of world-polymorphic functions. However we want the equality test on names to be accepted. Thus we at least need the relation to preserve name equalities across the relation in both directions. We require these relations to be functional and injective:

```

FunctionalAndInjective R =
  ∀ x₁ y₁ x₂ y₂ → R x₁ x₂ → R y₁ y₂
  → x₁ ≡ y₁ ↔ x₂ ≡ y₂

```

The relation `[[World]]` between two worlds α_1 and α_2 is the set of functional and injective relations between `Name` α_1 and `Name` α_2 . Then, two names x (in α_1) and y (in α_2) are related by `[[Name]]` α_r if and only if they are related by the α_r relation.

For the `_[[⊆]]_` relation, we exploit the fact there is only one way to use an inclusion witness, namely `coerceN`. Thus, for the purpose of building the model, we identify inclusions with their use in `coerceN`. Put differently, whatever the representation for inclusions is, the model takes them as functions from names to names. Yet another way to look at it is from the perspective of relation inclusions. A relation \mathcal{R}_1 is included in a relation \mathcal{R}_2 if and only if all pairs related by \mathcal{R}_1 are related by \mathcal{R}_2 as well. The `coerceN` function behaving like the identity function all these definitions coincide. For instance if we expand the definitions for `[[Name]]` and `[[→]]`, we get the definition for relation inclusion modulo the coercions:

```

_[[⊆]]_ αr βr p₁ p₂
= ∀ {x₁ x₂} → (x₁ , x₂) ∈ αr
  → (coerceN p₁ x₁ , coerceN p₂ x₂) ∈ βr

```

We now have to define operations on worlds that fit the logical relation. The case for `[[∅]]` is trivial. Then α_r `[[+1]]` is defined by $\{(x+1, y+1) \mid (x, y) \in \alpha_r\}$ and α_r `[[↑1]]` is defined by $\{(0, 0)\} \cup \alpha_r$ `[[+1]]`. We shown that both operations preserve functionality and injectiveness. Here are the signatures that these operations fit the relation:

```

record [[World]] α₁ α₂ : Set where
  constructor _,_
  field
    R : Name α₁ → Name α₂ → Set
    R-fun-inj : FunctionalAndInjective R

[[Name]] : ([[World]] [[→]] [[Set₀]]) Name Name
[[Name]] (R , _) x₁ x₂ = R x₁ x₂

_[[⊆]]_ : ([[World]] [[→]] [[World]] [[→]] [[Set₀]])
  _[[⊆]]_ _[[⊆]]_
_[[⊆]]_ αr βr α₁ ⊆ β₁ α₂ ⊆ β₂
= ([[Name]] αr [[→]] [[Name]] βr) (coerceN α₁ ⊆ β₁)
  (coerceN α₂ ⊆ β₂)

```

Figure 8. Relations for NAPA types

```

[[∅]] : [[World]] ∅ ∅
_[[+1]] : ([[World]] [[→]] [[World]]) _+1 _+1
_[[↑1]] : ([[World]] [[→]] [[World]]) _↑1 _↑1

```

4.3.2 NAPA values fit the relation

We now give a short overview of the proofs needed to show that our functions fit the relation. Thanks to the definition of `_[[↑1]]`, `zeroN` fits the relation by definition. Thanks to the definition of `_[[+1]]`, `addN 1` and `subtractN 1` fit the relation as well. These two are later extended to `addN k` and `subtractN k` by an induction on k . Since `[[−Name∅]]` receives names in the empty world, it trivially holds. Here is for instance the type signature for `[[addN]]`:

```

[[addN]] : (∀ (αr : [[World]] ) [[→]]
  (kr : [[N]] ) [[→]]
  [[Name]] αr [[→]]
  [[Name]] (αr [[+N]] kr)) addN addN

```

For `_[[==N]]_`, once unfolded, the statement tells that the equality test commutes with a renaming. This means that the result of the equality test does not change when its inputs are consistently renamed. The proof for `_[[==N]]_` is done in two parts. First, we have to relate the Boolean-valued function `==N` to the fact it decides equality on names. Second, we make use of the two properties (functionality and injectiveness) of the relation between the two worlds.

The definition of `cmpN` is not a simple induction on its first argument. It calls `_<=` (which does an induction) and returns a Boolean value. Based on this Boolean value, the function `cmpN` returns either `inj₁` or `inj₂` (the constructors of `_[[⊔]]_`) with the same name (with a different proof). To prove `_[[cmpN]]_` we show the equivalence with a simpler inductive function and show that this simpler function is in the relation. Thanks to the extensionality of the logical relation, no additional axiom is required to show that `cmpN` fits the relation.

Thanks to the definition of `_[[⊆]]_`, the proof that `coerceN` fits the relation is a simple application of the hypotheses. Then the real job is to show that all the inclusion rules fit the relation. This means that they all behave as the identity function. All eleven rules stated in figure 5 have been shown to fit the relation.

4.4 On the strength of free theorems

Every well-typed function comes with a free theorem [12]. However depending on the type of the function the strength of the theorem varies a lot. For instance at type $\text{Int} \rightarrow \text{Bool}$ it says no more than the function is deterministic. At type $\forall\{\mathbf{A} : \text{Set}\} \rightarrow \mathbf{A} \rightarrow \mathbf{A}$ it says that the function behaves like the identity function, which is much stronger. We now give a few elements of what can affect the strength of free theorems in our context. We will continue to use our Tm type to represent some data structures with names and binders but it could be any other.

Various term types The weakest type we can give to a value with names and binders is $\text{Tm} (\emptyset \uparrow \ell)$ for a given ℓ . Such a term can have a statically unknown number of distinct free variables, but we know that these variables are comprised in the interval $[0 .. \ell-1]$. The free theorem of $\forall\{\ell\} \rightarrow \text{Tm} (\emptyset \uparrow \ell) \rightarrow \text{Tm} (\emptyset \uparrow \ell)$ says no more than “the function is deterministic”. We also know because of typing that the resulting term cannot have ℓ -free occurrences.

A stronger type is to use any world instead of the empty world, but still have a known amount of bindings. A function of type $\forall\{\alpha \ell\} \rightarrow \text{Tm} (\alpha \uparrow \ell) \rightarrow \text{Tm} (\alpha \uparrow \ell)$ has a stronger free theorem. It says that the function commutes with a renaming of the variables in the world α (section 4.5). This is a common type to deal with open terms under a partial environment: $\forall\{\alpha \ell\} \rightarrow \text{Vec Info } \ell \rightarrow \text{Tm} (\alpha \uparrow \ell) \rightarrow \text{Tm} (\alpha \uparrow \ell)$.

The strongest type for terms is to make no assumptions on their free variables. This can be done by quantifying by an arbitrary world. The free theorem associated with the type $\forall\{\alpha\} \rightarrow \text{Tm } \alpha \rightarrow \text{Tm } \alpha$ says that the function commutes with renamings of the free variables. This particular type is studied in detail in the section 4.5.

Extra arguments Note that adding extra arguments to a function can drastically affect the strength of its free theorem. An extreme example is the type $\forall\{\alpha\} \rightarrow (\alpha \equiv \emptyset) \rightarrow \text{Tm } \alpha \rightarrow \text{Tm } \alpha$, which is ruined by its $\alpha \equiv \emptyset$ argument. While the above example is extreme, this is an important point to watch out for when adding extra arguments to a function. Another example is the type $\forall\{\alpha \beta\} \rightarrow \text{Tm } \alpha \rightarrow \text{Tm } \beta \rightarrow \text{Tm } \beta$ versus $\forall\{\alpha \beta\} \rightarrow (\text{Name } \alpha \rightarrow \text{Name } \beta \rightarrow \text{Bool}) \rightarrow \text{Tm } \alpha \rightarrow \text{Tm } \beta \rightarrow \text{Tm } \beta$, the first one cannot compare the free variables of the two given terms while the second can apply a user-supplied function to do so.

Shifting versus adding Last but not least using $+^w$ instead of \uparrow significantly improves the strength of the associated free theorem. Consider the function:

$$\begin{aligned} \text{protectedAdd} &: \forall\{\alpha\} \ell \mathbf{k} \rightarrow \text{Name } (\alpha \uparrow \ell) \\ &\quad \rightarrow \text{Name } (\alpha +^w \mathbf{k} \uparrow \ell) \\ \text{protectedAdd } \ell \mathbf{k} &= \text{protect}\uparrow \ell (\text{add}^w \mathbf{k}) \end{aligned}$$

Consider now a weaker type, namely:

$$\begin{aligned} \text{protectedAdd}\uparrow &: \forall\{\alpha\} \ell \mathbf{k} \rightarrow \text{Name } (\alpha \uparrow \ell) \\ &\quad \rightarrow \text{Name } (\alpha \uparrow \mathbf{k} \uparrow \ell) \\ \text{protectedAdd}\uparrow \ell \mathbf{k} &= \text{protect}\uparrow \ell (\text{add}\uparrow \mathbf{k}) \end{aligned}$$

We simply replaced the occurrence of $+^w$ by \uparrow . The consequences of this change are disastrous: this type allows more behaviors for its functions. Indeed the $\text{add}^w \mathbf{k}$ function can be given the latter type (but not the former) when using the appropriate inclusion witness to exploit the commutativity of \uparrow :

$$\begin{aligned} \text{unprotectedAdd} &: \forall\{\alpha\} \ell \mathbf{k} \rightarrow \text{Name } (\alpha \uparrow \ell) \\ &\quad \rightarrow \text{Name } (\alpha \uparrow \mathbf{k} \uparrow \ell) \\ \text{unprotectedAdd } \ell \mathbf{k} &= \text{coerce}^w (\subseteq\text{-exch-}\uparrow\text{-}\uparrow' \ell \mathbf{k}) \circ \text{add}\uparrow \mathbf{k} \end{aligned}$$

In our previous work [10] we had only the \uparrow operation. Our function `shiftName` (`protectedAdd` in this example) was defined with the `unprotectedAdd` behavior. It took us a long time to discover this mistake, since we thought that our logical relation argument was enough. In summary we want to emphasize that all the logical relations proofs have to be taken with great care. A weaker function type than expected can ruin the intended informal properties.

4.5 Using logical relations and parametricity

To formally show that a world-polymorphic function \mathbf{f} of type $\forall\{\alpha\} \rightarrow \text{Tm } \alpha \rightarrow \text{Tm } \alpha$ commutes with a renaming of the free variables, we proceed as follows. First we recall the natural definition of logical relation on the type Tm . Second we present the type Ren of renamings as injective functions. Third renaming is shown equivalent to being related at the type Tm . Finally we prove our commutation lemma by using the free-theorem associated to the function \mathbf{f} .

The logical relation for the type Tm is mechanical. It states that two terms are related if they have the same structure and related free variables:

$$\begin{aligned} \text{data } \llbracket \text{Tm} \rrbracket \{ \alpha_1 \alpha_2 \} (\alpha_r : \llbracket \text{World} \rrbracket \alpha_1 \alpha_2) : \\ &\quad \text{Tm } \alpha_1 \rightarrow \text{Tm } \alpha_2 \rightarrow \text{Set where} \\ \llbracket \mathbf{V} \rrbracket &: \forall\{\mathbf{x}_1 \mathbf{x}_2\} (\mathbf{x}_r : \llbracket \text{Name} \rrbracket \alpha_r \mathbf{x}_1 \mathbf{x}_2) \\ &\quad \rightarrow \llbracket \text{Tm} \rrbracket \alpha_r (\mathbf{V} \mathbf{x}_1) (\mathbf{V} \mathbf{x}_2) \\ \llbracket \llbracket \cdot \rrbracket \rrbracket &: \forall\{\mathbf{t}_1 \mathbf{t}_2 \mathbf{u}_1 \mathbf{u}_2\} \\ &\quad (\mathbf{t}_r : \llbracket \text{Tm} \rrbracket \alpha_r \mathbf{t}_1 \mathbf{t}_2) \\ &\quad (\mathbf{u}_r : \llbracket \text{Tm} \rrbracket \alpha_r \mathbf{u}_1 \mathbf{u}_2) \\ &\quad \rightarrow \llbracket \text{Tm} \rrbracket \alpha_r (\mathbf{t}_1 \cdot \mathbf{u}_1) (\mathbf{t}_2 \cdot \mathbf{u}_2) \\ \llbracket \lambda \rrbracket &: \forall\{\mathbf{t}_1 \mathbf{t}_2\} (\mathbf{t}_r : \llbracket \text{Tm} \rrbracket (\alpha_r \llbracket \uparrow 1 \rrbracket)) \mathbf{t}_1 \mathbf{t}_2 \\ &\quad \rightarrow \llbracket \text{Tm} \rrbracket \alpha_r (\lambda \mathbf{t}_1) (\lambda \mathbf{t}_2) \\ \llbracket \text{Let} \rrbracket &: \forall\{\mathbf{t}_1 \mathbf{t}_2 \mathbf{u}_1 \mathbf{u}_2\} \\ &\quad (\mathbf{t}_r : \llbracket \text{Tm} \rrbracket \alpha_r \mathbf{t}_1 \mathbf{t}_2) \\ &\quad (\mathbf{u}_r : \llbracket \text{Tm} \rrbracket (\alpha_r \llbracket \uparrow 1 \rrbracket)) \mathbf{u}_1 \mathbf{u}_2 \\ &\quad \rightarrow \llbracket \text{Tm} \rrbracket \alpha_r (\text{Let } \mathbf{t}_1 \mathbf{u}_1) (\text{Let } \mathbf{t}_2 \mathbf{u}_2) \end{aligned}$$

Then we need a notion of renaming. We choose to use injective functions over names. The type for a renaming is Ren , and the functions $\langle _ \rangle$ and $\llbracket _ \rrbracket$ respectively convert a renaming to a function over names, and to a relation over worlds:

$$\begin{aligned} \text{Ren} &: (\alpha \beta : \text{World}) \rightarrow \text{Set} \\ \langle _ \rangle &: \forall\{\alpha \beta\} \rightarrow \text{Ren } \alpha \beta \rightarrow \text{Tm } \alpha \rightarrow \text{Tm } \beta \\ \llbracket _ \rrbracket &: \forall\{\alpha \beta\} \rightarrow \text{Ren } \alpha \beta \rightarrow \llbracket \text{World} \rrbracket \alpha \beta \end{aligned}$$

We now observe that given a renaming Φ , it is equivalent for two terms \mathbf{t}_1 and \mathbf{t}_2 to be related by $\llbracket \text{Tm} \rrbracket \llbracket \Phi \rrbracket$ and for \mathbf{t}_2 to be equal to \mathbf{t}_1 renamed with Φ .

$$\begin{aligned} \llbracket \text{Tm} \rrbracket \llbracket \text{rename} \rrbracket : \\ &\quad \forall\{\alpha \beta\} (\Phi : \text{Ren } \alpha \beta) \{ \mathbf{t}_1 \mathbf{t}_2 \} \\ &\quad \rightarrow \llbracket \text{Tm} \rrbracket \llbracket \Phi \rrbracket \mathbf{t}_1 \mathbf{t}_2 \Leftrightarrow (\langle \Phi \rangle \mathbf{t}_1) \equiv \mathbf{t}_2 \end{aligned}$$

Finally given a function f and a proof f_r that f is in the logical relation, we can show that any renaming Φ commutes with the function f . To prove so we apply our $\llbracket \text{Tm} \rrbracket \Leftrightarrow \text{rename}$ lemma in both directions and use f_r with the renaming Φ lifted as a $\llbracket \text{World} \rrbracket$.

```

ren-comm :
  (f : ∀ {α} → Tm α → Tm α)
  (f_r : (∀ {α_r : [[World]]} [[→]] [[Tm]] α_r [[→]] [[Tm]] α_r) f f)
  → ∀ {α β} (Φ : Ren α β)
  → ⟨ Φ ⟩ ∘ f ≐ f ∘ ⟨ Φ ⟩
ren-comm f f_r Φ t
  = [[Tm]] ⇒ rename Φ
    (f_r ⟨ Φ ⟩ (rename ⇒ [[Tm]] Φ ≐ .refl))

```

5. Conclusion and future work

We have shown a new approach for a safer nameless programming interface. Our work relies on a different notion of worlds both finer and more abstract. The type of names while being represented as a natural number is kept abstract as well. Only a few functions are required from the interface to get complete expressiveness. Indeed while our worlds are more precise, nothing forces the programmer to be precise with them. Thus there is no loss of expressiveness compared to the `Fin` approach. Through concrete examples we have shown how we can program in this system using classical examples like capture avoiding substitution. However the most challenging result comes from the solid mechanized development we have made in AGDA. This development not only demonstrates the soundness of our approach but allows to derive properties of functions using the system. This way we have shown that world-polymorphic functions over terms commute with renamings of free variables. These soundness results are shown in a modular way and reuse the solid foundations of logical relations.

As future work we would like to explore more properties of this system. We have also seen that a major convenience problem in our system was to build world inclusion witnesses. We would like to address this problem through a witness inference system built within AGDA, using the recent reflection mechanism. Another time-consuming task was to apply the AGDA logical relations to types and operations, so we would like to explore the integration of inspection primitives and meta-programming facilities for AGDA, namely `TEMPLATE AGDA`. Finally we would like to investigate how this system could scale to representations of well-typed terms.

Acknowledgements Thanks to François Pottier, Jean-Philippe Bernardy, Alexandre Pilkiewicz, and the anonymous reviewers for providing us with very valuable feedback.

References

- [1] Thorsten Altenkirch. A formalization of the strong normalization proof for System F in LEGO. In J.F. Groote M. Bezem, editor, *Typed Lambda Calculi and Applications*, LNCS 664, pages 13 – 28, 1993.
- [2] Thorsten Altenkirch and Bernhard Reus. Monadic presentations of lambda terms using generalized inductive types. In *Computer Science Logic*, volume 1683 of *Lecture Notes in Computer Science*, pages 453–468. Springer, 1999.
- [3] Françoise Bellegarde and James Hook. Substitution: A formal methods case study using monads and transformations. *Sci. Comput. Program.*, 23(2-3):287–311, 1994.
- [4] Jean-Philippe Bernardy, Patrik Jansson, and Ross Paterson. Parametricity and dependent types. In *Proceedings of the 15th ACM SIGPLAN international conference on Functional programming, ICFP '10*, pages 345–356, New York, NY, USA, 2010. ACM.
- [5] Richard Bird and Ross Paterson. de Bruijn notation as a nested datatype. *Journal of Functional Programming*, 9(1):77–91, January 1999.
- [6] Nicolaas G. de Bruijn. Lambda-calculus notation with nameless dummies: a tool for automatic formula manipulation with application to the Church-Rosser theorem. *Indag. Math.*, 34(5):381–392, 1972.
- [7] Conor McBride and James McKinna. The view from the left. *J. Funct. Program.*, 14:69–111, January 2004.
- [8] Conor McBride and Ross Paterson. Applicative programming with effects. *Journal of Functional Programming*, 18(1):1–13, 2008.
- [9] Nicolas Pouillard. Nameless, painless (Agda code), 2011. <http://tiny.nicolaspouillard.fr/NaPa.agda>.
- [10] Nicolas Pouillard and François Pottier. A fresh look at programming with names and binders. In *Proceedings of the 15th ACM SIGPLAN international conference on Functional programming, ICFP '10*, pages 217–228, New York, NY, USA, September 2010. ACM.
- [11] John C. Reynolds. Types, abstraction and parametric polymorphism. In *Information Processing 83*, pages 513–523. Elsevier Science, 1983.
- [12] Philip Wadler. Theorems for free! In *Conference on Functional Programming Languages and Computer Architecture (FPCA)*, pages 347–359, September 1989.